

* Stata Analysis Examples Replication for ASDA 3rd Edition
* Berglund Fall 2024

*** CHAPTER 7

* Section 7.5 Application

* Table 7.1

```
use "P:\ASDA3\Data Sets for Analysis Examples and Stata R Code\nhanes1112.dta", clear
svyset sdmvpsu [pweight = WTMEC2YR], strata(sdmvstra) ///
vce(linearized) singleunit(missing)
```

```
gen bpxdil_1 = BPXD11
replace bpxdil_1 = . if BPXD11 == 0
```

```
gen age18p = 1 if age >= 18 & age != .
replace age18p = 0 if age < 18
```

```
svy, subpop(age18p): mean age
gen agec = age - 46.36
```

```
svy, subpop(age18p): regress bpxdil_1 i.RIDRETH1
test 2.RIDRETH1 3.RIDRETH1 4.RIDRETH1 5.RIDRETH1
svy, subpop(age18p): regress bpxdil_1 i.marcat
test 2.marcat 3.marcat
svy, subpop(age18p): regress bpxdil_1 i.riagendr
test 2.riagendr
svy, subpop(age18p): regress bpxdil_1 agec
test agec
```

```
regress bpxdil_1 i.RIDRETH1 i.riagendr agec if age18p == 1
```

* Table 7.2 Unweighted OLS model

```
regress bpxdil_1 i.RIDRETH1 i.riagendr i.marcat agec if age18p == 1
```

* Table 7.3 WLS OLS model

```
regress bpxdil_1 i.RIDRETH1 i.riagendr i.marcat agec ///
if age18p == 1 [pweight = WTMEC2YR]
```

* Table 7.4 Design Based model

```
svyset sdmvpsu [pweight = WTMEC2YR], strata(sdmvstra) ///
vce(linearized) singleunit(missing)
```

```
svy, subpop(age18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec
```

```
estat effects, deff
```

* Specify all design features, but not the survey weights

```
svyset sdmvpsu, strata(sdmvstra)
```

* Add the weights and all two-way interactions w/the weights

* Table 7.5 Design Based Approach

```
svy, subpop(age18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec ///
WTMEC2YR c.WTMEC2YR#i.RIDRETH1 c.WTMEC2YR#i.riagendr ///
c.WTMEC2YR#i.marcat c.WTMEC2YR#c.agec
```

* Test whether the weights are informative

```
test WTMEC2YR 2.RIDRETH1#c.WTMEC2YR 3.RIDRETH1#c.WTMEC2YR 4.RIDRETH1#c.WTMEC2YR ///
5.RIDRETH1#c.WTMEC2YR 2.riagendr#c.WTMEC2YR 2.marcat#c.WTMEC2YR ///
3.marcat#c.WTMEC2YR c.agec#c.WTMEC2YR
```

* Diagnostics

```
svyset sdmvpsu [pweight = WTMEC2YR], strata(sdmvstra) ///
vce(linearized) singleunit(missing)
```

```
svy, subpop(age18p): regress bpxdil_1 i.RIDRETH1 ///
```

```

i.riagendr i.marcat agec

predict resids, resid
scatter resids agec

graph save "Graph" "P:\ASDA3\Replication Stata\Chapter 7\Figure73a.gph"

* Generate agesq
gen agesq = agec * agec

* Table 7.5
svy, subpop(ager18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec agesq

estat effects

predict resids2, resid
scatter resids2 agec

graph save "Graph" "P:\ASDA3\Replication Stata\Chapter 7\Figure73b.gph"

graph combine "P:\ASDA3\Replication Stata\Chapter 7\Figure73a.gph" "P:\ASDA3\Replication
Stata\Chapter 7\Figure73b.gph"

* Testing interactions
svy, subpop(ager18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec agesq ///
i.RIDRETH1#c.agec i.RIDRETH1#c.agesq

test 2.RIDRETH1#c.agec 3.RIDRETH1#c.agec ///
4.RIDRETH1#c.agec 5.RIDRETH1#c.agec 2.RIDRETH1#c.agesq ///
3.RIDRETH1#c.agesq 4.RIDRETH1#c.agesq 5.RIDRETH1#c.agesq

svy, subpop(ager18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec agesq ///
i.RIDRETH1#c.agec i.RIDRETH1#c.agesq ///
i.riagendr#c.agec i.riagendr#c.agesq

test 2.riagendr#c.agec 2.riagendr#c.agesq

svy, subpop(ager18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec c.agec#c.agec ///
i.RIDRETH1#c.agec i.RIDRETH1#c.agec#c.agec ///
i.riagendr#c.agec i.riagendr#c.agec#c.agec

margins RIDRETH1, at(agec=(-30(5)30))
marginplot
graph save "Graph" "P:\ASDA3\Replication Stata\Chapter 7\fig74a.gph"

margins riagendr, at(agec=(-30(5)30))
marginplot
graph save "Graph" "P:\ASDA3\Replication Stata\Chapter 7\fig74b.gph"
graph combine "P:\ASDA3\Replication Stata\Chapter 7\fig74a.gph" "P:\ASDA3\Replication Stata\Chapter
7\fig74b.gph"

* More diagnostics
* Table 7.6
svy, subpop(ager18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcat agec c.agec#c.agec ///
i.RIDRETH1#c.agec i.RIDRETH1#c.agec#c.agec ///
i.riagendr#c.agec i.riagendr#c.agec#c.agec

predict ehat1, resid

symplot ehat1, name(sym_ehat1_1, replace) ///
title(Symplot of Residuals)

histogram ehat1, normal name(h_ehat1, replace) ///
title(Histogram of Residuals)

```

```

qnorm ehat1, name(qnorm_ehat1, replace) ///
title(Normal Q-Q Plot of Residuals)

predict yhat1, xb

scatter ehat1 yhat1, name(ehat1xyhat1, replace) ///
title(Residuals vs. Predicted Y)

* Figure 7.5
graph combine sym_ehat1_1 h_ehat1 qnorm_ehat1 ///
ehat1xyhat1, rows(2)

* Table 7.6
* q-weighted approach
regress WTMEC2YR i.RIDRETH1 i.riagendr i.marcatec agec ///
c.agec#c.agec i.RIDRETH1#c.agec i.RIDRETH1#c.agec#c.agec ///
i.riagendr#c.agec i.riagendr#c.agec#c.agec if age18p == 1

predict w_hat if age18p == 1, xb
summarize w_hat, detail
replace w_hat = 4809 if w_hat < 0

gen q_WTMEC2YR = WTMEC2YR / w_hat
svyset sdmvpsu [pweight = q_WTMEC2YR], strata(sdmvstra)

svy, subpop(age18p): regress bpxdil_1 i.RIDRETH1 ///
i.riagendr i.marcatec agec c.agec#c.agec ///
i.RIDRETH1#c.agec i.RIDRETH1#c.agec#c.agec ///
i.riagendr#c.agec i.riagendr#c.agec#c.agec

```